
;-------------------------------------------------------------------

;                Creep of Cylindrical Cavern

;                anisotropic in-situ stress field

;                (units: m, MPa, year)

;-------------------------------------------------------------------

config creep

;

; create model (quarter-symmetry)

gen zone radcyl size 8 12 6 12  rat 1.0 1.0 1.0 1.2  fill  &

    p2 0 0 0  p4 250 0 0  p5 0 250 0  p7 250 250 0  &

    p0 0 0 45  p1 250 0 45  p3 0 250 45  p6 250 250 45  &

    p10 45 0 0  p11 0 45 0  &

    p8 45 0 45  p9 0 45 45

;

gen zone radcyl size 8 8 6 12  rat 1.0 1.2 1.0 1.2  fill  &

    p0 0 0 45  p3 250 0 45  p1 0 250 45  p6 250 250 45  &

    p2 0 0 250  p5 250 0 250  p4 0 250 250  p7 250 250 250  &

    p10 0 45 250  p11 45 0 250  &

    p8 0 45 45  p9 45 0 45

;

; define four groups

group salt   range z 0 30

group cavern range cyl end1 0 0 0 end2 0 0 15 rad 45

group mat2   range z 30 45

group mat3   range z 45 250

;

; assign material models

model power   range group cavern

model power   range group salt

model elastic range group mat2

model elastic range group mat3

;

; material properties

; salt: E=10e6 psi ; Poisson's ratio=0.3

prop bulk 58.3e3  shear 26.9e3 range group salt

prop bulk 58.3e3  shear 26.9e3 range group cavern

; A = 3.9 MPa**(-n) * yr**(-1) ; n=4.9

prop a_1=3.9e-7 n_1=4.9  range group salt

prop a_1=3.9e-7 n_1=4.9  range group cavern

;

; mat2: E=30e6 psi ; Poisson's ratio=0.3

prop bulk 175e3   shear 80.7e3 range group mat2

;

; mat3: E=10e6 psi ; Poisson's ratio=0.3



prop bulk 58.3e3  shear 26.9e3 range group mat3

;

; boundary and initial conditions

fix x  ran x -0.1 0.1

fix y  ran y -0.1 0.1

fix z  ran z -0.1 0.1

fix x  range x 249 251

fix y  range y 249 251

apply  szz -35 range z 249 251

ini    sxx -56  syy -28  szz -35

;

; creep parameters

set creep on

set cr dt 1.0e-5

set cr mindt 1.0e-5

set cr maxdt 1.0e-3

set cr dt auto on

;

; histories

hist unbal

hist crtime

hist dt

;

hist gp xdis 45 0 0

hist gp ydis 0 45 0

hist gp zdis 0 0 15

;

hist z sxx  0 0 16

hist z syy  0 0 16

hist z szz  0 0 16

;

hist z sxx  0 0 31

hist z syy  0 0 31

hist z szz  0 0 31

;

; solve for pre-excavation state of stress

solve age 4.0

save cavern_1.sav

;

; excavate cavern

model null   range group cavern

;

; gradually reduce stress at excavation

def loadhist



  while_stepping

  tt = 0.10

  if crtime <= tt

    xloadhist = -35 + 28 * crtime / tt

    yloadhist = -35 + 28 * crtime / tt

    zloadhist = -35 + 28 * crtime / tt

  else

    xloadhist = -7

    yloadhist = -7

    zloadhist = -7

  end_if

end

;

apply sxx 1 his xloadhist range cyl end1 0 0 -0.1 end2 0 0 15.1 rad 45.1

apply syy 1 his yloadhist range cyl end1 0 0 -0.1 end2 0 0 15.1 rad 45.1

apply szz 1 his zloadhist range cyl end1 0 0 -0.1 end2 0 0 15.1 rad 45.1

;

; histories

hist delete

hist unbal

hist crtime

hist dt

;

hist gp xdis 45 0 0

hist gp ydis 0 45 0

hist gp zdis 0 0 15

;

hist xloadhist

hist yloadhist

hist zloadhist

; solve for 0.1 yr excavation

ini xdis=0 ydis=0 zdis=0

set creep time=0.0

set creep dt=1.0e-5

set creep dt auto on

solve age 0.1

save cavern_2.sav

;

; solve for 1 yr of creep

solve age 1.0

save cavern_3.sav

;

ret

